

## 1. Sistemas dinámicos unidimensionales (Práctica 1)

**Ejercicio 1.1.** El siguiente procedimiento de Maple calcula los primeros términos de la órbita de un punto  $p$  para una función  $f$ . Tiene por argumentos:

- $f$  = función que queremos analizar.
  - $p$  = punto inicial.
  - $n$  = número de puntos de la órbita de  $p$ .
- ```
> orbitanumerica:=proc(f,p,n)
> local a,k;
> a:=evalf(p);
> lprint(cat('a',1)=a);
> for k from 1 to n do;
>   a:=f(a);
>   lprint(cat('a',k+1)=a);
> od:
> end:
```

Aplicar este procedimiento con  $f = x^2$ ,  $p = 0.9$  y  $n = 6$ .

Modificando ligeramente este procedimiento obtenemos otro que nos calcule la órbita de un punto entre el término  $m$ -ésimo y el  $n$ -ésimo. Tiene por argumentos:

- $f$  = función que queremos analizar.
  - $p$  = punto inicial.
  - $m$  = índice inicial de la porción que se muestra la órbita de  $p$  (ha de ser  $m > 1$ ).
  - $n$  = índice final de la porción que se muestra la órbita de  $p$ .
- ```
> orbitanumericaavanzada:=proc(f,p,m,n)
> local a,k;
> a:=evalf(p);
> for k from 1 to m-2 do;
>   a:=f(a);
> od:
> for k from 1 to n-m+1 do;
>   a:=f(a);
>   lprint(cat('a',m-1+k)=a);
> od:
> end:
```

Aplicar este procedimiento a  $f = x^2 - 1$ ,  $p = 1.2$ ,  $m = 101$  y  $n = 105$ . ¿Hacia donde converge la órbita de  $p$ ?

**Ejercicio 1.2.** Escribir el procedimiento de Maple que hace el análisis gráfico de una función  $f$  a partir de un punto  $p$ . Tiene por argumentos:

- $f$  = función que queremos analizar.
- $p$  = punto inicial.
- $n$  = número de puntos de la órbita de  $p$ .
- $d$  = extremo izquierdo del intervalo de visualización.
- $e$  = extremo derecho del intervalo de visualización.

```

> analisisgrafico:=proc(f,p,n,d,e)
> local p0,p1,p2,a,l,k,b,p3:
> p0:=plot([[p,0]],d..e,d..e,style=point,symbol=BOX,color=red):
> p1:=plot(f,d..e,d..e,color=blue):
> p2:=plot(x,x=d..e,d..e,color=black,linestyle=4):
> a:=evalf(p):
> l:=[[a,0]]:
> for k from 1 to n-1 do:
>   b:=f(a): l:=[op(l),[a,b],[b,b]]: a:=b:
> od:
> b:=f(a): l:=[op(l),[a,b]]:
> p3:=plot(l,d..e,d..e,style=line,color=red):
> plots[display]({p0,p1,p2,p3},scaling=constrained);
> end:

```

Se pide:

- Aplicar este procedimiento con  $f = x^2$ ,  $p = 0.9$  y  $n = 6$ .
- A partir del procedimiento anterior obtener un procedimiento para cada uno de los casos siguientes:
  - Que nos muestre la órbita evolucionando en el tiempo.
  - Que nos muestre la órbita estática de varios puntos.
  - Que nos muestre la órbita de varios puntos evolucionando en el tiempo.

Ninguno de los procedimientos habrá de tener mas de 6 líneas. Por ejemplo, el segundo de ellos tiene la siguiente estructura:

```

> analisisgraficovariospuntos:=proc(f,lp,n,d,e)
> local i:
> plots[display]([seq(analisisgrafico(f,lp[i],n,d,e),i=1..nops(lp))],
>               view=[d..e,d..e],scaling=constrained);
> end:

```

donde lp es una lista de puntos.

El primero sería parecido al segundo con la opción `insequence=true` en el comando `plots[display]`.

El tercero será una mezcla de los otros dos.

- Aplicar el segundo de los procedimientos anteriores a  $f = x^2$  en el intervalo  $[-2, 2]$  para estudiar el comportamiento global del sistema.

**Ejercicio 1.3.** Obtener, modificando el procedimiento `analisisgrafico`, otro procedimiento que nos muestre un estado avanzado de la órbita, no representando la parte inicial de la misma. Aplicarlo a  $f = x^2 - 1.3$ ,  $p = 1.5$ ,  $m = 101$  y  $n = 105$ . ¿Hacia donde converge la órbita de  $p$ ?

**Ejercicio 1.4.** Estudiar, utilizando aquellos de los procedimientos anteriores que creas conveniente, el comportamiento de los sistemas dinámicos definidos por  $f(x) = x^3 - 0.9x^2 - 0.9x$  y  $f(x) = x^3 - 1.1x^2 - 1.1x$ , esto es, puntos fijos y periódicos y cuencas de atracción. Utilizar los comandos de maple “diff” para derivar y “solve” para resolver ecuaciones (o “fsolve” para soluciones aproximadas), para obtener con precisión las coordenadas de los puntos fijos y periódicos que encuentres y confirmar su carácter atractivo o repulsivo.

**Ejercicio 1.5 (Opcional).** Hacer un procedimiento que busque ciclos atractivos (hasta cierto orden) en una función